

Deformations of Matrix Factorisations with Macaulay2

BRADFORD HOVINEN

ABSTRACT. We describe a *Macaulay2* package which computes versal deformations of matrix factorisations, or equivalently maximal Cohen-Macaulay modules on hypersurfaces.

1. INTRODUCTION. We present here a *Macaulay2* [M2] package, called `ModuleDeformations`, which computes versal deformations of matrix factorisations of polynomials. We begin with some notation and definitions.

Definition 1.1. Let S be a regular ring and $f \in S$ a regular element. A *matrix factorisation* of f is a pair (A, B) of square matrices of the same size over S such that $AB = f \cdot I = BA$, where I is the identity matrix of the same dimension.

Matrix factorisations of f are equivalent to maximal Cohen-Macaulay (MCM) modules on the hypersurface defined by f : given a matrix factorisation (A, B) of f , the cokernel of A is an MCM module over $S/(f)$, and all MCM modules over $S/(f)$ arise in this way; see [E].

For the course of this paper, we work with germs of varieties whose distinguished point is denoted by 0 . All morphisms $X \rightarrow Y$ map $0 \in X$ to $0 \in Y$. Our package represents a germ as the coordinate ring of an affine variety of which the germ is the localisation (or completion) at the ideal generated by the variables of the representing ring. It is an error if that ideal is not prime (e.g., if it is in fact the whole ring). With this in mind, we now define deformations of MCM modules.

Definition 1.2. Let X be a germ of a hypersurface and M_0 a maximal Cohen-Macaulay module over X . Let $Y \rightarrow \Sigma$ be a flat map of germs whose fibre over $0 \in \Sigma$ is isomorphic to X . A *deformation of M_0 in the deformation theory of $Y \rightarrow \Sigma$* is a pair (S, \mathcal{M}) with S a germ of a variety equipped with a map $S \rightarrow \Sigma$ and \mathcal{M} a module over $S \times_{\Sigma} Y$, flat over S , whose restriction to $0 \times_{\Sigma} X \subseteq S \times_{\Sigma} Y$ is isomorphic to M_0 . In the sequel, we will suppress from our notation the map $Y \rightarrow \Sigma$ when it is understood from context, referring simply to a *deformation of M_0* .

A deformation (T, \mathcal{M}') is *induced from* (S, \mathcal{M}) if there exists a map $\phi : T \rightarrow S$ commuting with the maps $T \rightarrow \Sigma$ and $S \rightarrow \Sigma$ and such that $\mathcal{M}' \cong (\phi \times_{\Sigma} \text{id}_Y)^* \mathcal{M}$.

The deformation (S, \mathcal{M}) is *versal* if any deformation of M_0 is induced from (S, \mathcal{M}) .

There are two cases which we discuss. In the *absolute* case, Σ is a reduced point, and the module M_0 is deformed over Y only. In the *relative* case, Σ is the base space of a nontrivial deformation of X and Y is the total space of the deformation. In this case, M_0 is deformed also along the given deformation of X .

`ModuleDeformations` is an implementation of the *Massey Product Algorithm* for constructing versal deformations of maximal Cohen-Macaulay modules over hypersurfaces. The algorithm is described

2010 *Mathematics Subject Classification.* 14B12, 13C14.

ModuleDeformations version 1.0.

in the absolute case in [S] and in general in [H]. The implementation has been designed to take advantage of graded structure whenever possible. If the input is quasihomogeneous, then the output will be too, and the computation may run much more efficiently.

Below, we present three examples of the use of `ModuleDeformations`: one example in the absolute case, one simple example in the relative case, and one more complicated example in the relative case.

2. A DEFORMATION IN THE ABSOLUTE CASE. The following *Macaulay2* session illustrates the use of this package to construct a versal deformation of the maximal ideal (x, y) of $\mathbb{Q}[[x, y]]/(x^4 - y^3)$, which is maximal Cohen-Macaulay of rank one. First, we load the package and define the ring $\mathbb{Q}[x, y]/(x^4 - y^3)$ in which we compute. This ring is graded if we set $\deg x = 3$ and $\deg y = 4$.

```
i1 : loadPackage "ModuleDeformations";
i2 : R = QQ[x, y, Degrees => {3, 4}] / (x^4 - y^3);
```

We now deform the module. The output is a pair (\mathcal{O}_S, M) , where S is the base space of the deformation and M is a module defined over the *ambient* space of $S \times Y$ whose restriction to $S \times Y$ is the deformation module. The procedure does not substitute the module M to the ring of $S \times Y$ because the resulting computation of a Gröbner basis of the defining ideal of $\mathcal{O}_{S \times Y}$ is in certain cases too expensive.

```
i3 : (OS, M) = deformMCMModule module ideal (x, y);
i4 : describe OS
      QQ[xi_1, xi_2]
      1      2
o4 = -----
      3      4
      (- xi_1 - xi_2, 0)
      1      2

i5 : M
o5 = cokernel {3} | y+xi_1 x^3-x^2xi_2+xxi_2^2-xi_2^3 |
      {4} | -x-xi_2 -y^2+yxi_1-xi_1^2 |

o5 : QQ[x, y, xi_1, xi_2]-module, quotient of (QQ[x, y, xi_1, xi_2])^2
      1      2
      1      2
```

We may compare the presentation matrix of M with that of the original module.

```
i6 : presentation module ideal (x, y)
o6 = {3} | y x^3 |
      {4} | -x -y^2 |
      2      2
o6 : Matrix R <--- R
```

We see that the latter is obtained from the former by setting the deformation parameters ξ_1 and ξ_2 to 0, as expected.

The maximal ideal has the same deformation theory as the residue field (see [H, §4.5]). Thus the base space S of its semiuniversal deformation is just the Hilbert scheme of one point, which is isomorphic to the original singularity Y . The substitution $\xi_1 \mapsto y$, $\xi_2 \mapsto x$ in the above output demonstrates this. The module is isomorphic to the defining ideal of the diagonal embedding of Y into $Y \times S \cong Y \times Y$.

3. A DEFORMATION IN THE RELATIVE CASE. In the following example we consider the Whitney umbrella $Y = \text{Spec } \mathbb{Q}[[x, y, z]]/(z^2 - (y-x)y^2)$. Reduction modulo x gives the A_2 singularity $\text{Spec } \mathbb{Q}[[y, z]]/(z^2 - y^3)$. We deform the maximal ideal (y, z) of the A_2 singularity in the deformation theory defined by the projection $\phi : Y \rightarrow \Sigma := \text{Spec } \mathbb{Q}[[x]]$.

First we define the rings \mathcal{O}_Σ , \mathcal{O}_Y and the ring homomorphism $\phi^* : \mathcal{O}_\Sigma \rightarrow \mathcal{O}_Y$ corresponding to ϕ . Setting $\deg x := 2$, $\deg y := 2$, and $\deg z := 3$ gives the associated affine rings a graded structure, which ϕ^* preserves.

```
i7 : OSigma = QQ[x, Degrees => {2}];
i8 : OY = QQ[y,z,x, Degrees => {2,3,2}] / (z^2-(y-x)*y^2);
i9 : phi = map(OY, OSigma, matrix {{x}});
o9 : RingMap OY <--- OSigma
```

We then construct the ring \mathcal{O}_X as the coordinate ring of the fibre over 0 of the map $Y \rightarrow \Sigma$. Over \mathcal{O}_X we will construct the module M_0 to be deformed.

```
i10 : OX = trim (OY / phi ideal OSigma_0);
```

We now deform the maximal ideal (y, z) of \mathcal{O}_X . The procedure returns the pair (\mathcal{O}_S, M) , where \mathcal{O}_S is the ring of the base space and M is a module defined over the ambient ring of $S \times_\Sigma Y$ whose restriction to $S \times_\Sigma Y$ is the deformation module. The procedure does not substitute the module M into the ring of $S \times_\Sigma Y$ for the same reason as in the absolute case above.

```
i11 : (S,M) = deformMCMModule(module ideal (y,z), phi);
i12 : describe S
      QQ[x, xi_1, xi_2]
           1      2
o12 = -----
      2      2      3
      (xi_1 + x*xi_1 + xi_2, 0)
           1      2      2
i13 : M
o13 = cokernel {2} | z+xi_1 y2-yx-yxi_2+xxi_2+xi_2^2 |
                {3} | -y-xi_2 -z+xi_1 |
                                2
o13 : QQ[y, z, x, xi_1, xi_2]-module, quotient of (QQ[y, z, x, xi_1, xi_2])
           1      2                                1      2
```

Similar to the example of §2, the base space S of the deformation is just the Hilbert scheme of one point on the Whitney umbrella Y . The substitution $y \mapsto -\xi_2, z \mapsto \xi_1$ establishes this isomorphism in the output above. The module M is again isomorphic to the defining ideal of the diagonal embedding of Y in $Y \times_\Sigma S \cong Y \times_\Sigma Y$.

4. A MORE COMPLICATED EXAMPLE. In the following example, we compute a versal deformation of the normalisation of the E_6 curve singularity onto the discriminant in a semiuniversal deformation of the A_3 singularity. A semiuniversal deformation of the A_3 singularity $\text{Spec } \mathbb{C}[[x, y]]/(x^4 - y^2)$ is given by the equation $x^4 - y^2 + 6a_2x^2 + 4a_3x + 3a_4 = 0$, with base space $\text{Spec } \mathbb{C}[[a_2, a_3, a_4]]$. The assignment of degree i to each parameter a_i , along with $\deg x = 1$ and $\deg y = 2$, makes the equation quasihomogeneous. The discriminant is a hypersurface in $\text{Spec } \mathbb{C}[[a_2, a_3, a_4]]$ defined by the equation

$$a_4^3 - a_3^4 + 6a_2a_3^2a_4 - 6a_2^2a_4^2 - 2a_2^3a_3^2 + 9a_2^4a_4 = 0.$$

which is quasihomogeneous of degree 12.

The E_6 singularity $a_4^3 - a_3^4 = 0$ is obtained by intersection with the hyperplane $a_2 = 0$. The *Macaulay2* code below constructs the ring R of the discriminant, the map $\text{Spec } R \rightarrow \Sigma \cong \text{Spec } \mathbb{C}[[a_2]]$, and the ring $\tilde{R} \cong \text{Spec } \mathbb{C}[[a_3, a_4]]/(a_4^3 - a_3^4)$ of the E_6 singularity.

```
i14 : OSigma = QQ[a_2, Degrees => {2}];
i15 : R = QQ[a_2, a_3, a_4, Degrees => {2, 3, 4}] /
      ideal(a_4^3 - a_3^4 + 6*a_2*a_3^2*a_4 - 6*a_2^2*a_4^2 - 2*a_2^3*a_3^2 + 9*a_2^4*a_4);
i16 : phi = map(R, OSigma, matrix {{a_2}});
o16 : RingMap R <--- OSigma
i17 : Rtilde = R / phi ideal OSigma_0;
```

The normalisation $N(\tilde{R})$ of \tilde{R} , obtained by the parametrisation $\tilde{R} \rightarrow \mathbb{C}[[t]]$ given by $a_3 \mapsto t^3$, $a_4 \mapsto t^4$, has presentation matrix

$$\begin{bmatrix} a_4 & 0 & -a_3^2 \\ -a_3 & a_4 & 0 \\ 0 & -a_3 & a_4 \end{bmatrix}.$$

As modules over \tilde{R} , $N(\tilde{R})$ is isomorphic to the *conductor ideal* $(a_3^2, a_3 a_4, a_4^2) = (a_3, a_4)^2$ of the normalisation map. Thus we construct and deform the module:

```
i18 : (S, M) = deformMCMModule(module (ideal (a_3, a_4))^2, phi);
```

We may then investigate S :

```
i19 : ambient S
o19 = QQ[a_2, xi_1, xi_2, xi_3, xi_4, xi_5, xi_6]
      2   1   2   3   4   5   6
o19 : PolynomialRing
i20 : degrees S
o20 = {{2}, {5}, {4}, {4}, {3}, {3}, {2}}
o20 : List
i21 : dim S
o21 = 4
i22 : numgens trim ideal S
o22 = 3
```

There are six parameters ξ_1, \dots, ξ_6 with degrees $\deg \xi_1 = 5$, $\deg \xi_2 = 4 = \deg \xi_3$, $\deg \xi_4 = 3 = \deg \xi_5$, and $\deg \xi_6 = 2$. The base space is a complete intersection of dimension 4 in a seven-dimensional ambient space.

Setting a_2 to 0 in the above output recovers a versal deformation of the normalisation of \tilde{R} in the absolute case. Such a deformation is among those constructed in [S], and indeed the results therein may be recovered using this *Macaulay2* module. A direct comparison requires applying to the base space of the deformation a change of variables which can be done by (somewhat tedious) inspection.

REFERENCES.

- [E] D. Eisenbud, *Homological algebra on a complete intersection, with an application to group representations*, Trans. Amer. Math. Soc. **260** (1980), no. 1, 35 – 64.
- [M2] D.R. Grayson and M.E. Stillman, *Macaulay 2, a software system for research in algebraic geometry*, available at www.math.uiuc.edu/Macaulay2/.
- [H] B. Hovinen, *Matrix factorizations of the classical discriminant*, PhD thesis, University of Toronto, 2008.

[S] A. Siqueland, *The method of computing formal moduli*, J. Algebra **241** (2001), no. 1, 292 – 327.

RECEIVED : 2008-11-09

REVISED : 2009-12-11

ACCEPTED : 2010-04-17

hovinen@math.uni-hannover.de : Institut für Algebraische Geometrie, Leibniz Universität Hannover, Welfengarten 1,
30167 Hannover, Germany.